

- 1. IDENTIFICATION
- 1.1 Digital-8-6-U-Sym
- 1.2 Octal Memory Dump
- 1.3 March 17, 1965

2. ABSTRACT

This routine will read the console switches twice to obtain the upper and lower limits of an area of memory, then type on the Teletype an absolute address plus the octal contents of the first four words specified and repeat this until the block is exhausted, at which time the user may repeat the operation.

3. REQUIREMENTS

3.1 Storage

This routine requires 77 core locations.

3.3 Equipment

Basic PDP-8 with ASR 33.

4. USAGE

4.1 Loading

This routine is loaded with the Binary Loader. (See Digital-8-2-U-Rim for a complete discussion of the Binary Loader and its use.)

4.2 Calling Sequence

None. It cannot be called as a subroutine.

4.3 Switch Settings

The switch register is used to enter the starting address. It is also used to enter two delimiting addresses which define the area of memory to be typed.

4.4 Start Up/Entry

Assuming the program is in memory, proceed as follows:

4.4.1 Set the starting address in the switch register and press the LOAD ADDRESS key.

4.4.2 Set the low address into the switch register and press the START key.

4.4.3 The computer will halt. Set the high address into the switch register and press the CONTINUE key.

4.4.4 Timeout will occur and the computer will halt when the designated block has been listed.

4.4.5 To type another block, set the low address into the switch register and press the CONTINUE key.

4.4.6 Repeat step 4.4.3 above.

6. DESCRIPTION

6.1 Discussion

Reference to the Flow Charts (Section 11.1) will illustrate the following discussion.

Three inner subroutines are used. These are tagged TYPN, CRLF, and PNUM. Respectively, these subroutines cause a digit to be typed, cause a carriage return followed by a line feed to be typed, and control the typeout of a 4-digit octal number, i.e., either a memory address or data stored in a single word in memory. PNUM is illustrated with a flow chart. The other two subroutines are composed of a few instructions only.

Main routine flow is as follows. After initialization, a carriage return line feed is typed. Next, the starting address followed by several spaces is typed. A loop is next entered to type the contents of four memory locations (if four remain to be typed). If during the latter loop the routine finds it has processed the last memory location, the loop exits, a carriage return line feed is typed, and a JMP to ODUM executed.

If after typing the contents of four successive addresses there remains more data to be typed, a JMP to DUM2 repeats the process.

6.2 Example

The Octal Dump routine was used to dump itself as follows:

<u>Address</u>		<u>Four Data Words</u>		
7400	7200	7404	3256	7402
7404	7404	7040	1256	3257
7410	5213	2260	5226	1265
7414	3260	4247	1256	4661
7420	1265	3247	1264	4241
7424	2247	5222	7200	1656
7430	4661	1264	4241	2256
7434	2257	5211	4247	7402
7440	5200	7433	6046	6041
7444	5243	7200	5641	0000
7450	7200	1263	4241	1262
7454	4241	5647	7456	7741
7460	7774	7466	0212	0215
7464	0240	7774	7431	3311
7470	1265	3310	1311	7004
7474	7004	7006	3311	1311
7500	0313	1314	4712	1311
7504	2310	5274	7200	5666
7510	0000	0000	7441	0007
7514	0260			

8. FORMAT

8.1 External Data

See the example in Section 6.2.

9. EXECUTION TIME

This routine is entirely output limited.

10. PROGRAM

10.4 Program Listing

		*7400		
7400	7200	ODUM,	CLA	
7401	7404		OSR	/GET LOWER LIMIT
7402	3256		DCA LOCK	
7403	7402		HLT	
7404	7404		OSR	/GET UPPER LIMIT
7405	7040		CMA	
7406	1256		TAD LOCK	
7407	3257		DCA LIM	/INITIALIZE RANGE COUNTER
7410	5213		JMP .+3	
7411	2260	DUM2,	ISZ LPCN	/END OF LINE
7412	5226		JMP DUM3	
7413	1265		TAD CONS+3	/-4
7414	3260		DCA LPCN	/RESET ITEM COUNTER
7415	4247		JMS CRLF	/CARR. RET. AND LINE FEED
7416	1256		TAD LOCK	
7417	4661		JMS I PRIN	/INTER-COM. TO PNUM
7420	1265		TAD CONS+3	/-4
7421	3247		DCA CRLF	
7422	1264		TAD CONS+2	/240 (SPACE)
7423	4241		JMS TYPN	
7424	2247		ISZ CRLF	
7425	5222		JMP .-3	
7426	7200	DUM3,	CLA	
7427	1656		TAD I LOCK	
7430	4661		JMS I PRIN	/INTER-COM. TO PNUM
7431	1264		TAD CONS+2	/240 (SPACE)
7432	4241		JMS TYPN	
7433	2256		ISZ LOCK	/INDEX LOCATION POINTER


7434	2257		ISZ LIM	/END OF RANGE
7435	5211		JMP DUM2	
7436	4247		JMS CRLF	
7437	7402		HLT	
7440	5200		JMP ODUM	/GO AGAIN
7441	0000	TYPN,	0	/TYPE CHARACTER SUBROUTINE
7442	6046		TLS	
7443	6041		TSF	
7444	5243		JMP .-1	
7445	7200		CLA	
7446	5641		JMP I TYPN	
7447	0000	CRLF,	0	/CR AND LF SUBROUTINE
7450	7200		CLA	
7451	1263		TAD CONS+1	/215 (CARR. RET.)
7452	4241		JMS TYPN	
7453	1262		TAD CONS	/212 (LINE FEED)
7454	4241		JMS TYPN	
7455	5647		JMP I CRLF	
7456	0000	LOCK,	0	
7457	0000	LIM,	0	
7460	0000	LPCN,	0	
7461	7466	PRIN,	PNUM	/INTER-COM. TO PNUM
7462	0212	CONS,	212	/CONSTANTS
7463	0215		215	
7464	0240		240	
7465	7774		7774	
7466	0000	PNUM,	0	/PRINT NUMBER SUBROUTINE
7467	3311		DCA PTEM	
7470	1265		TAD CONS+3	/-4
7471	3310		DCA DCN	/INITIALIZE DIGIT COUNTER
7472	1311		TAD PTEM	
7473	7004		RAL	
7474	7004	PNU2,	RAL	
7475	7006		RTL	
7476	3311		DCA PTEM	
7477	1311		TAD PTEM	
7500	0313		AND PCON	/7
7501	1314		TAD PCON+1	/260
7502	4712		JMS I TDIT	/TYPN (TYPE A DIGIT)
7503	1311		TAD PTEM	
7504	2310		ISZ DCN	
7505	5274		JMP PNU2	

7506	7200		CLA	
7507	5666		JMP I PNUM	
7510	0000	DCN,	0	/DIGIT COUNTER
7511	0000	PTEM,	0	
7512	7441	TDIT,	TYPN	
7513	0007		7	/CONSTANTS
7514	0260		260	
CONS	7462			
CRLF	7447			
DCN	7510			
DUM2	7411			
DUM3	7426			
LIM	7457			
LOCK	7456			
LPCN	7460			
ODUM	7400			
PCON	7513			
PNUM	7466			
PNU2	7474			
PRIN	7461			
PTEM	7511			
TDIT	7512			
TYPN	7441			

11. DIAGRAMS

11.1 Flow Charts

11.1.1 Main Flow



11.1.2 PNUM Subroutine

